

NAG Library Routine Document

D01FCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

D01FCF attempts to evaluate a multidimensional integral (up to 15 dimensions), with constant and finite limits, to a specified relative accuracy, using an adaptive subdivision strategy.

2 Specification

```

SUBROUTINE D01FCF (NDIM, A, B, MINPTS, MAXPTS, FUNCTN, EPS, ACC, LENWRK,      &
                  WRKSTR, FINVAL, IFAIL)
INTEGER          NDIM, MINPTS, MAXPTS, LENWRK, IFAIL
REAL (KIND=nag_wp) A(NDIM), B(NDIM), FUNCTN, EPS, ACC, WRKSTR(LLENWRK),    &
                  FINVAL
EXTERNAL        FUNCTN

```

3 Description

D01FCF returns an estimate of a multidimensional integral over a hyper-rectangle (i.e., with constant limits), and also an estimate of the relative error. You set the relative accuracy required, return values for the integrand via a routine argument `FUNCTN`, and also set the minimum and maximum acceptable number of calls to `FUNCTN` (in `MINPTS` and `MAXPTS`).

The routine operates by repeated subdivision of the hyper-rectangular region into smaller hyper-rectangles. In each subregion, the integral is estimated using a seventh-degree rule, and an error estimate is obtained by comparison with a fifth-degree rule which uses a subset of the same points. The fourth differences of the integrand along each coordinate axis are evaluated, and the subregion is marked for possible future subdivision in half along that coordinate axis which has the largest absolute fourth difference.

If the estimated errors, totalled over the subregions, exceed the requested relative error (or if fewer than `MINPTS` calls to `FUNCTN` have been made), further subdivision is necessary, and is performed on the subregion with the largest estimated error, that subregion being halved along the appropriate coordinate axis.

The routine will fail if the requested relative error level has not been attained by the time `MAXPTS` calls to `FUNCTN` have been made; or, if the amount `LENWRK` of working storage is insufficient. A formula for the recommended value of `LENWRK` is given in Section 5. If a smaller value is used, and is exhausted in the course of execution, the routine switches to a less efficient mode of operation; only if this mode also breaks down is insufficient storage reported.

D01FCF is based on the `HALF` subroutine developed by van Dooren and de Ridder (1976). It uses a different basic rule, described in Genz and Malik (1980).

4 References

Genz A C and Malik A A (1980) An adaptive algorithm for numerical integration over an N-dimensional rectangular region *J. Comput. Appl. Math.* **6** 295–302

van Dooren P and de Ridder L (1976) An adaptive algorithm for numerical integration over an N-dimensional cube *J. Comput. Appl. Math.* **2** 207–217

5 Arguments

- 1: NDIM – INTEGER *Input*
On entry: n , the number of dimensions of the integral.
Constraint: $2 \leq \text{NDIM} \leq 15$.
- 2: A(NDIM) – REAL (KIND=nag_wp) array *Input*
On entry: the lower limits of integration, a_i , for $i = 1, 2, \dots, n$.
- 3: B(NDIM) – REAL (KIND=nag_wp) array *Input*
On entry: the upper limits of integration, b_i , for $i = 1, 2, \dots, n$.
- 4: MINPTS – INTEGER *Input/Output*
On entry: must be set to the minimum number of integrand evaluations to be allowed.
On exit: contains the actual number of integrand evaluations used by D01FCF.
- 5: MAXPTS – INTEGER *Input*
On entry: the maximum number of integrand evaluations to be allowed.
Constraints:
 $\text{MAXPTS} \geq \text{MINPTS}$;
 $\text{MAXPTS} \geq \alpha$, where $\alpha = 2^{\text{NDIM}} + 2 \times \text{NDIM}^2 + 2 \times \text{NDIM} + 1$.
- 6: FUNCTN – REAL (KIND=nag_wp) FUNCTION, supplied by the user. *External Procedure*
 FUNCTN must return the value of the integrand f at a given point.

The specification of FUNCTN is:

```
FUNCTION FUNCTN (NDIM, Z)
REAL (KIND=nag_wp) FUNCTN
INTEGER NDIM
REAL (KIND=nag_wp) Z(NDIM)
```

- 1: NDIM – INTEGER *Input*
On entry: n , the number of dimensions of the integral.
- 2: Z(NDIM) – REAL (KIND=nag_wp) array *Input*
On entry: the coordinates of the point at which the integrand f must be evaluated.

FUNCTN must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub)program from which D01FCF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

- 7: EPS – REAL (KIND=nag_wp) *Input*
On entry: the relative error acceptable to you. When the solution is zero or very small relative accuracy may not be achievable but you may still set EPS to a reasonable value and check for the error exit IFAIL = 2.
Constraint: $\text{EPS} > 0.0$.
- 8: ACC – REAL (KIND=nag_wp) *Output*
On exit: the estimated relative error in FINVAL.

- 9: LENWRK – INTEGER *Input*
On entry: the dimension of the array WRKSTR as declared in the (sub)program from which D01FCF is called.
Suggested value: for maximum efficiency, $\text{LENWRK} \geq (\text{NDIM} + 2) \times (1 + \text{MAXPTS}/\alpha)$ (see argument MAXPTS for α).
 If LENWRK is less than this, D01FCF will usually run less efficiently and may fail.
Constraint: $\text{LENWRK} \geq 2 \times \text{NDIM} + 4$.
- 10: WRKSTR(LenWRk) – REAL (KIND=nag_wp) array *Workspace*
- 11: FINVAL – REAL (KIND=nag_wp) *Output*
On exit: the best estimate obtained for the integral.
- 12: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if $\text{IFAIL} \neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\text{IFAIL} = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: D01FCF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NDIM < 2,
 or NDIM > 15,
 or MAXPTS is too small,
 or $\text{LENWRK} < 2 \times \text{NDIM} + 4$,
 or $\text{EPS} \leq 0.0$.

IFAIL = 2

MAXPTS was too small to obtain the required relative accuracy EPS. On soft failure, FINVAL and ACC contain estimates of the integral and the relative error, but ACC will be greater than EPS.

IFAIL = 3

LENWRK was too small. On soft failure, FINVAL and ACC contain estimates of the integral and the relative error, but ACC will be greater than EPS.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

A relative error estimate is output through the argument ACC.

8 Parallelism and Performance

D01FCF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

D01FCF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

Execution time will usually be dominated by the time taken to evaluate FUNCTN, and hence the maximum time that could be taken will be proportional to MAXPTS.

10 Example

This example estimates the integral

$$\int_0^1 \int_0^1 \int_0^1 \int_0^1 \frac{4z_1 z_3^2 \exp(2z_1 z_3)}{(1+z_2+z_4)^2} dz_4 dz_3 dz_2 dz_1 = 0.575364.$$

The accuracy requested is one part in 10000.

10.1 Program Text

```
! D01FCF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d01fcfe_mod

! D01FCF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp
! .. Implicit None Statement ..
Implicit None
! .. Accessibility Statements ..
Private
Public                                :: functn
! .. Parameters ..
Integer, Parameter, Public            :: ndim = 4, nout = 6
Integer, Parameter, Public            :: maxpts = 1000*ndim
Integer, Parameter, Public            :: lenwrk = (ndim+2)*(1+maxpts/(2**ndim &
+2*ndim*ndim+2*ndim+1))
```

```

Contains
  Function functn(ndim,z)

!      .. Function Return Value ..
      Real (Kind=nag_wp)          :: functn
!      .. Scalar Arguments ..
      Integer, Intent (In)        :: ndim
!      .. Array Arguments ..
      Real (Kind=nag_wp), Intent (In) :: z(ndim)
!      .. Intrinsic Procedures ..
      Intrinsic                   :: exp
!      .. Executable Statements ..
      functn = 4.0E0_nag_wp*z(1)*z(3)*z(3)*exp(2.0E0_nag_wp*z(1)*z(3))/      &
              (1.0E0_nag_wp+z(2)+z(4))**2

      Return

  End Function functn
End Module d01fcfe_mod
Program d01fcfe

!      D01FCF Example Main Program

!      .. Use Statements ..
      Use nag_library, Only: d01fcf, nag_wp
      Use d01fcfe_mod, Only: functn, lenwrk, maxpts, ndim, nout
!      .. Implicit None Statement ..
      Implicit None
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: acc, eps, finval
      Integer                    :: ifail, minpts
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: a(:), b(:), wrkstr(:)
!      .. Executable Statements ..
      Write (nout,*) 'D01FCF Example Program Results'

      Allocate (a(ndim),b(ndim),wrkstr(lenwrk))

      a(1:ndim) = 0.0E0_nag_wp
      b(1:ndim) = 1.0E0_nag_wp
      eps = 0.0001E0_nag_wp
      minpts = 0

      ifail = -1
      Call d01fcf(ndim,a,b,minpts,maxpts,functn,eps,acc,lenwrk,wrkstr,finval, &
                 ifail)

      Select Case (ifail)
      Case (0,2:)
        Write (nout,*)
        Write (nout,99999) 'Requested accuracy = ', eps
        Write (nout,99998) 'Estimated value = ', finval
        Write (nout,99999) 'Estimated accuracy = ', acc
      End Select

99999 Format (1X,A,E12.2)
99998 Format (1X,A,F12.4)
      End Program d01fcfe

```

10.2 Program Data

None.

10.3 Program Results

D01FCF Example Program Results

Requested accuracy =	0.10E-03
Estimated value =	0.5754
Estimated accuracy =	0.99E-04
